算數運算（減法二）

上一篇文章我們提到了 8 bits 的減法運算，如果要進行 Unsigned 32 bits 的減法運算，我們可以利用下面這個範例常式的寫法。

```
DATA_SUBTRACT
      MOV     R2,#04H
      CLR     C
$1    MOV     A,@R0
      SUBB    A,@R1
      MOV     @R0,A
      INC     R0
      INC     R1
      DJNZ    R2,$1
      RET
```

和加法一樣，我們實際推導一下它運作的情況，這樣有助於對程式的了解。首先假設 R0=10H、R1=14H，那麼這個範例便是將 13H、12H、11H、10H 這些位址的值減去 17H、16H、15H、14H 這些位址的值，再假設在 13H　10H 的值分別為 03H、02H、01H、00H，而在 17H　14H 的值分別為 02H、03H、01H、00H。此時的 R2 是用來做為迴圈數的控制。

次 數	C	R1				R0			
		(17H)	(16H)	(15H)	(14H)	(13H)	(12H)	(11H)	(10H)
1. R2=04H R1=14H R0=10H	0	02H	03H	01H	00H	03H	02H	01H	00H
2. R2=03H R1=15H R0=11H	0	02H	03H	01H	00H	03H	02H	00H	00H
3. R2=02H R1=16H R0=12H	1	02H	03H	01H	00H	03H	FFH	00H	00H
4. R2=01H R1=17H R0=13H	0	02H	03H	01H	00H	00H	FFH	00H	00H

表格中標示紅色與藍色的部分，是每次運算時更動到的部分，當系統進行最低 Byte 的減法運算時，先將 (10H) 中的 00H 搬到 Acc 累加器中，再把 Acc 累加器的值減去 (14H) 中的 00H 與旗標值，而此時的 CY 值為啟始的預設值 0，因此相減後所得到的值 00H - 00H - 0 = 00H 會再存回 Acc 累加器中，最後把 Acc 累加器的值再搬到 (10H) 的位址裡。此時的減法因為沒有任何的借位狀況，所以 CY 旗標值還是 0。

旗威科技有限公司　地址：高雄市三民區昌裕街 18-1 號　網址：http://www.chipware.com.tw
技術專線：07-395-5152　技術支援傳真：07-395-5155　E-mail：chipware@chipware.com.tw

第二個 Byte 的相減同第一個 Byte 一樣，由於相減的結果最後是要存回 (11H) 裡，所以 (11H) 的值便成了 00H。

第三個 Byte 的相減有些不同，02H - 03H - 0 = FFH 產生了借位，因此 CY 旗標會設定成 1，但這裡的 1 並不會在這一次的運算裡減去，因為 SUBB 運算時是利用前一次運算所得到的 CY 值進行運算的。此時的 (12H) 會被填為 FFH。

第四個 Byte 的相減因為沒有借位，所以 CY 值會被再設定成 0，然而在運算減法時會將前一次的 CY 值減去，所以 (13H) 最後所得到的值是 03H - 02H - 1 = 00H。

在這個常式範例裡，有幾個必須要特別注意的要點：第一、R0、R1 及 R2 為運算位址的暫存區及迴圈數的設定值，必須避開運算的位址用到這三個暫存器。第二、R0 與 R1 內所設定的四個位址不可有重疊的情形，否則會出現很嚴重的運算錯誤。第三、如果相減的數值不是 32 bits，而是 24 bits 或是 48 bits，只需將 R2 的值改成 3 或 6，並確定 R0 及 R1 所設定的位址範圍沒有重疊即可。

在此筆者要提出一個問題，還記得我們提過的利用減數反相與被減數相加來達到減法的功能嗎？如果要應用在 Unsigned 32 bits 的減法運算，您會怎麼做呢？底下這一段範例便是利用該法完成的，試著自己推導看看，並比較這兩種方法的差異吧！
（標示紅色的部分是程式的關鍵唷！）

```
DATA_SUBTRACT_2
      MOV     R2,#04H
      SETB    C
$1    MOV     A,@R1
      CPL     A
      ADDC    A,@R0
      MOV     @R0,A
      INC     R0
      INC     R1
      DJNZ    R2,$1
      RET
```


旗威科技有限公司　地址：高雄市三民區昌裕街 18-1 號　網址：http://www.chipware.com.tw
技術專線：07-395-5152　技術支援傳真：07-395-5155　E-mail：chipware@chipware.com.tw